



Wikidata Query Service (WDQS)

Envel Le Hir – CC BY-SA 4.0

www.lehir.net

3 janvier 2020

France, Rennes, Les Champs Libres,
Musée de Bretagne



Programme

- Tour de table
- Présentation de Wikidata
- Présentation de SPARQL
- Utilisation du Wikidata Query Service

Wikidata

- Projet-frère de l'encyclopédie Wikipédia, lancé en octobre 2012
- Base de connaissance
 - base de données sur plus de 72 millions de concepts en janvier 2020
 - libre : licence Creative Commons « Transfert dans le Domaine Public » CC0 (≈ domaine public)
 - collaborative
 - multilingue

Représentation d'un concept

The image shows a Wikidata concept page for "Douglas Adams (Q42)". The page is annotated with labels on the left and right sides, pointing to various elements. The main content includes a title, a description, a list of statements (Déclarations), and a list of references (références).

Annotations:

- libellé:** Points to the title "Douglas Adams (Q42)".
- description:** Points to the text "écrivain anglais de science-fiction" and "Douglas Noel Adams | Douglas Noel Adams".
- propriété:** Points to the "scolarité" property in the statements section.
- rang:** Points to the "2 références" section.
- groupe de déclarations:** Points to the entire "Déclarations" section.
- identifiant de l'élément:** Points to "(Q42)".
- alias:** Points to "Douglas Noel Adams | Douglas Noel Adams".
- valeur:** Points to "St John's College" in the first statement.
- qualificatifs:** Points to the table of qualifications for "St John's College".
- références dépliées:** Points to the expanded reference for "Encyclopædia Britannica Online".
- références repliées:** Points to the collapsed reference for "Brentwood School".

Main Content:

Titre: Douglas Adams (Q42)

Description: écrivain anglais de science-fiction
Douglas Noel Adams | Douglas Noel Adams
Plus de langues

Déclarations

scolarité

St John's College	
date de fin	1974
principale spécialité d'étude	littérature anglaise
diplôme	baccalauréat en arts
date de début	1971

2 références

références dépliées:

affirmé dans	Encyclopædia Britannica Online
URL de la référence	http://www.ndb.com/people/731/000023662/
langue originale	anglais
date de consultation	7 décembre 2013
éditeur	NNDB
titre	Douglas Adams (anglais)

+ ajouter une référence

références repliées:

Brentwood School

date de fin	1970
date de début	1959

0 références

+ ajouter


Ontologies

- Wikidata est organisé en projets
- Exemples :
 - projet [France](#) et ses sous-projets Communes, Monuments historiques, Élus, etc.
 - projet [Equines](#)
- Permet de :
 - discuter avec les autres contributeurs intéressés par le sujet
 - établir les propriétés à utiliser pour chaque type de concept
 - lister des requêtes de maintenance, directement ou avec des outils ([Listeria](#), [Integraality](#), etc.)

SPARQL

- « SPARQL Protocol and RDF Query Language »
 - standard du W3C : norme SPARQL 1.1 (2013) ([overview](#), [query language](#))
- Langage de requête de base de données graphe RDF
- Triplet RDF : sujet → prédicat → objet
- Exemples :
 - Les Misérables → date de publication → 1862
 - Les Misérables → auteur → Victor Hugo
 - Victor Hugo → prénom → Victor
 - Victor Hugo → nom de famille → Hugo
 - Victor Hugo → date de naissance → 26 février 1802

Modèle de données RDF



Wikidata Query Service (WDQS)

- Adresse : <https://query.wikidata.org/>
- Permet d'interroger Wikidata grâce à SPARQL
 - afficher les données
 - les visualiser sous différentes formes
 - les télécharger
 - les partager
- Raccourcis
- Lag :
<https://grafana.wikimedia.org/d/000000489/wikidata-query-service?orgId=1&fullscreen&panelId=8>

Première requête

Liste des chevaux dans Wikidata :

```
SELECT ?horse  
WHERE {  
    ?horse wdt:P31 wd:Q726 .  
}
```

Libellés et descriptions

Liste des chevaux avec leurs libellés et descriptions :

```
SELECT ?horse ?horseLabel ?horseDescription
WHERE {
    ?horse wdt:P31 wd:Q726 .
    SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
```

Exhaustivité

Liste complète des chevaux dans Wikidata :

```
SELECT ?horse ?horseLabel ?horseDescription
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 .
    SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
```

Exhaustivité : attention aux doublons

Liste complète des chevaux dans Wikidata (sans doublon) :

```
SELECT DISTINCT ?horse ?horseLabel ?horseDescription
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 .
    SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
```

Afficher des images

Liste des chevaux avec une image :

```
#defaultView: ImageGrid
```

```
SELECT ?horse ?horseLabel ?image
```

```
WHERE {
```

```
    ?horse wdt:P31/wdt:P279* wd:Q726 .
```

```
    ?horse wdt:P18 ?image .
```

```
    SERVICE wikibase:label { bd:serviceParam wikibase:language  
"[AUTO_LANGUAGE],fr" . }
```

```
}
```

Afficher sur une carte

Carte des lieux de naissance des chevaux :

```
#defaultView:Map
```

```
SELECT ?horse ?horseLabel ?coord
```

```
WHERE {
```

```
    ?horse wdt:P31/wdt:P279* wd:Q726 .
```

```
    ?horse wdt:P19/wdt:P625 ?coord .
```

```
    SERVICE wikibase:label { bd:serviceParam wikibase:language
```

```
    "[AUTO_LANGUAGE],fr" . }
```

```
}
```

Condition simple

Liste des juments :

```
SELECT ?horse ?horseLabel ?horseDescription
WHERE {
  ?horse wdt:P31/wdt:P279* wd:Q726 .
  ?horse wdt:P21 wd:Q43445 .
  SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
```

Condition simple (forme condensée)

Liste des juments :

```
SELECT ?horse ?horseLabel ?horseDescription
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 ; wdt:P21 wd:Q43445 .
    SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
```


Limiter le nombre de résultats

Liste de 100 juments :

```
SELECT ?horse ?horseLabel ?horseDescription
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 .
    ?horse wdt:P21 wd:Q43445 .
    SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
LIMIT 100
```

Récupérer des valeurs

Liste des chevaux avec leurs dates de naissance :

```
SELECT ?horse ?horseLabel ?horseDescription ?birthdate
WHERE {
  ?horse wdt:P31/wdt:P279* wd:Q726 .
  ?horse wdt:P569 ?birthdate .
  SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
```

Récupérer optionnellement des valeurs

Liste des chevaux avec leurs dates de naissance si elles existent :

```
SELECT ?horse ?horseLabel ?horseDescription ?birthdate
WHERE {
  ?horse wdt:P31/wdt:P279* wd:Q726 .
  OPTIONAL { ?horse wdt:P569 ?birthdate . }
  SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
```

Récupérer les valeurs manquantes

Liste des chevaux sans date de naissance :

```
SELECT ?horse ?horseLabel ?horseDescription
WHERE {
  ?horse wdt:P31/wdt:P279* wd:Q726 .
  FILTER NOT EXISTS { ?horse wdt:P569 [] . }
  SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
```

Trier les résultats (ordre croissant)

Liste des chevaux triés par leurs dates de naissance :

```
SELECT ?horse ?horseLabel ?horseDescription ?birthdate
WHERE {
  ?horse wdt:P31/wdt:P279* wd:Q726 .
  ?horse wdt:P569 ?birthdate .
  SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
ORDER BY ?birthdate
```

Trier les résultats (ordre décroissant)

Liste des chevaux triés par leurs dates de naissance :

```
SELECT ?horse ?horseLabel ?horseDescription ?birthdate
WHERE {
  ?horse wdt:P31/wdt:P279* wd:Q726 .
  ?horse wdt:P569 ?birthdate .
  SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
ORDER BY DESC(?birthdate)
```

Limiter le nombre de résultats

Liste des 100 chevaux avec les plus anciennes dates de naissance :

```
SELECT ?horse ?horseLabel ?horseDescription ?birthdate
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 .
    ?horse wdt:P569 ?birthdate .
    SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
ORDER BY ?birthdate
LIMIT 100
```

Les dates

- RDF stocke des dates complètes, sans tenir compte de leur précision réelle connue dans Wikidata
- Pour obtenir un résultat exact, il est nécessaire de :
 - récupérer toutes les dates
 - filtrer pour ne conserver que le « meilleur » rang
 - filtrer pour ne conserver que la précision souhaitée :
 - jour : 11
 - mois : 10
 - année : 9
 - décade : 8
 - etc.
 - fonctions : `DAY ()` , `MONTH ()` , `YEAR ()`

Les dates : filtrer par précision (jour)

Liste des chevaux avec une date de naissance connue au jour près :

```
SELECT ?horse ?horseLabel ?horseDescription ?birthdate
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 .
    ?horse p:P569 [ rdf:type wikibase:BestRank ; psv:P569 [
wikibase:timeValue ?birthdate ; wikibase:timePrecision
?birthdate_precision ] ] .
    FILTER(?birthdate_precision >= 11)
}
ORDER BY ?birthdate
```

Les dates : filtrer par précision (année)

Liste des chevaux avec une date de naissance connue à l'année près :

```
SELECT ?horse ?horseLabel ?horseDescription ?year
WHERE {
  ?horse wdt:P31/wdt:P279* wd:Q726 .
  ?horse p:P569 [ rdf:type wikibase:BestRank ; psv:P569 [
wikibase:timeValue ?birthdate ; wikibase:timePrecision
?birthdate_precision ] ] .
  FILTER(?birthdate_precision >= 9)
  BIND(YEAR(?birthdate) AS ?year)
}
ORDER BY ?year
```

Les dates : condition

Liste des chevaux avec une date de naissance postérieure à 2010 :

```
SELECT ?horse ?horseLabel ?horseDescription ?year
WHERE {
  ?horse wdt:P31/wdt:P279* wd:Q726 .
  ?horse p:P569 [ rdf:type wikibase:BestRank ; psv:P569 [
wikibase:timeValue ?birthdate ; wikibase:timePrecision
?birthdate_precision ] ] .
  FILTER(?birthdate_precision >= 9)
  FILTER(?birthdate >= "2010-01-01"^^xsd:dateTime) .
  BIND(YEAR(?birthdate) AS ?year)
}
```

Les dates : condition (variante)

Liste des chevaux avec une date de naissance postérieure à 2010 :

```
SELECT ?horse ?horseLabel ?horseDescription ?year
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 .
    ?horse p:P569 [ rdf:type wikibase:BestRank ; psv:P569 [
wikibase:timeValue ?birthdate ; wikibase:timePrecision
?birthdate_precision ] ] .
    FILTER(?birthdate_precision >= 9)
    BIND(YEAR(?birthdate) AS ?year)
    FILTER(?year >= 2010) .
}
```

Compter le nombre de résultats

Nombre de chevaux dans Wikidata :

```
SELECT (COUNT(*) AS ?count)
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 .
}
```

```
SELECT (COUNT(DISTINCT ?horse) AS ?count)
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 .
}
```

Aggréger

Nombre de chevaux par race de cheval :

```
SELECT ?race ?raceLabel (COUNT(DISTINCT ?horse) AS ?count)
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 .
    ?horse wdt:P4743 ?race .
    SERVICE wikibase:label { bd:serviceParam wikibase:language
"[AUTO_LANGUAGE],fr" . }
}
GROUP BY ?race ?raceLabel
ORDER BY ?raceLabel
```

Aggréger

Longévité minimum et maximum par race de cheval :

```
SELECT ?race ?raceLabel (MIN(?age) AS ?age_min) (MAX(?age)
AS ?age_max)
WHERE {
    ?horse wdt:P31/wdt:P279* wd:Q726 ; wdt:P4743 ?race .
    ?horse wdt:P569 ?birthdate ; wdt:P570 ?deathdate .
    BIND(?deathdate - ?birthdate AS ?age)
}
GROUP BY ?race ?raceLabel
ORDER BY ?raceLabel
```

Questions ?